

Problem 1a

```

1 fun all_except_option (s,xs) =
2   case xs of
3     [] => NONE
4   | x::xs' => if same_string(s,x)
5                 then SOME xs'
6                 else case all_except_option(s,xs') of
7                     NONE => NONE
8                     | SOME y => SOME(x::y)
9

```

- It is fine for a solution to be more complicated if the reason is that it allows `s` to be in `xs` multiple times although the problem does not require handling this situation.
- It is fine to use the `=` operator directly instead of `same_string` even though the specification asked for `same_string`.
- A helper function is really not needed here, but many people will probably use one. Generally consider a good such solution to be worth a 4, but a 5 is okay if it is really nice.

Remember that you are grading on general style, not how close to the sample solution a student solution is. It is perfectly fine for a solution to be significantly different from the sample, as long as it has good style.

Problem 1b

Here is a sample solution:

```

1 fun get_substitutions1 (substitutions,str) =
2   case substitutions of
3     [] => []
4   | x::xs => case all_except_option(str,x) of
5               NONE => get_substitutions1(xs,str)
6               | SOME y => y @ get_substitutions1(xs,str)
7

```

- Give at most a 3 if they use helper functions other than `all_except_option` and ML's append operator `@`. This includes functions defined in `get_substitutions1` as they are not helpful here.
- If a solution uses a local `val` binding for no useful purpose, still give a 5 if everything else is great, but this really is inferior style. For example:

```

1 fun get_substitutions1 (substitutions,str) =
2   case substitutions of
3     [] => []
4   | x::xs => let val foo = all_except_option(str,x)
5                 in case foo of
6                     NONE => get_substitutions1(xs,str)
7                     | SOME y => y @ get_substitutions1(xs,str)
8                 end

```

Remember that you are grading on general style, not how close to the sample solution a student solution is. It is perfectly fine for a solution to be significantly different from the sample, as long as it has good style.

Problem 1c

Here is a sample solution:

```
1 fun get_substitutions2 (substitutions, str) =  
2   let fun loop (acc, substs_left) =  
3     case substs_left of  
4       [] => acc  
5       | x::xs => loop ((case all_except_option(str, x) of  
6                           NONE => acc  
7                           | SOME y => acc @ y),  
8                           xs)  
9   in  
10    loop ([], substitutions)  
11  end
```

In addition to general style, the most important thing we are checking is that there is a tail recursive local helper function.

- Give at most a 2 if they do not define a helper function.
- Give at most a 3 if their helper function is not defined locally (inside `get_substitutions2`).
- Give at most a 3 if the helper function is not tail-recursive: where it calls itself (only one place is needed), there should be no more work afterward. So the result of the recursive call should not be an argument to anything like another function, `@`, etc.
- Give at most a 3 if you cannot easily find the initial call to the helper function with `[]` for an accumulator argument.

The sample solution uses a case-expression directly for the accumulator argument to the recursive call loop. This is not necessary for good style. A fine alternative is something like:

```
1 case all_except_option(str, x) of  
2   NONE => loop(acc, xs)  
3   | SOME y => loop(acc @ y, xs)
```

It is also okay if a solution uses local variables in good style for computing the values that are then passed to the recursive call.

Remember that you are grading on general style, not how close to the sample solution a student solution is. It is perfectly fine for a solution to be significantly different from the sample, as long as it has good style.

Problem 1d

Here is a sample solution:

```

1 fun similar_names (substitutions,name) =
2   let
3     val {first=f, middle=m, last=l} = name
4     fun make_names xs =
5       case xs of
6         [] => []
7         | x::xs' => {first=x, middle=m, last=l}::(make_names(xs'))
8   in
9     name::make_names(get_substitutions2(substitutions,f))
10  end

```

- It is fine, in fact even slightly better, to have the record pattern in the function argument, something like `fun similar_names (substitutions,{first=f, middle=m, last=l}) =`
- Give at most a 4 if they do not use one of the `get_substitutions` functions defined earlier.
- Give at most a 3 for a solution longer than 25 lines, not including comments. (Exact line count is never important, but this is over 2.5 times as long as the sample solution.)

Remember that you are grading on general style, not how close to the sample solution a student solution is. It is perfectly fine for a solution to be significantly different from the sample, as long as it has good style.

Problem 2a

Here is a sample solution:

```

1 fun card_color card =
2   case card of
3     (Clubs,_) => Black
4   | (Diamonds,_) => Red
5   | (Hearts,_) => Red
6   | (Spades,_) => Black

```

- Do not penalize solutions that do not use the wildcard pattern (something like `(Clubs,x)` for each pattern).
- Do not penalize solutions that take an argument that is a pair pattern (something like `fun card_color (suit,value) =`).
- Do not penalize solutions that use function-pattern syntax, something like:

```

1 fun card_color (Clubs,_) = Black
2   | card_color (Diamonds,_) = Red ...

```

- Give a 4 for this solution, which does not use nested patterns:

```

1 fun card_color card =
2   let val (s,v) = card
3   in case s of
4     Clubs => Black
5     | Diamonds => Red
6     | Hearts => Red
7     | Spades => Black
8   end

```

Remember that you are grading on general style, not how close to the sample solution a student solution is. It is perfectly fine for a solution to be significantly different from the sample, as long as it has good style.

Problem 2b

Here is a sample solution:

```
1 fun card_value card =
2   case card of
3     (_,Jack) => 10
4   | (_,Queen) => 10
5   | (_,King) => 10
6   | (_,Ace) => 11
7   | (_,Num n) => n
```

Follow similar guidelines as for the previous problem.

Problem 2c

Here are two sample solutions:

```
1 fun remove_card (cs,c,e) =
2   case cs of
3     [] => raise e
4   | x::cs' => if x = c then cs' else x :: remove_card(cs',c,e)
5
6 fun remove_card (cs,c,e) =
7   let fun f cs =
8         case cs of
9           [] => raise e
10          | x::cs' => if x = c then cs' else x :: f cs'
11       in
12         f cs
13     end
```

- There are unlikely to be too many ways to make solutions much more complicated without introducing poor style.
- Do not penalize solutions that define their own function to see if two cards are equal rather than using the = operator.

Remember that you are grading them on general style, not how close to the sample solution their solution is. It is perfectly fine for their solution to be significantly different from the sample, as long as it has good style.

Problem 2d

Here are two sample solutions:

```

1 fun all_same_color cs =
2   case cs of
3     [] => true
4   | [_] => true
5   | head::neck::tail => card_color head = card_color neck
6     andalso all_same_color(neck::tail)
7
8 fun all_same_color cs =
9   case cs of
10     head::neck::tail => card_color head = card_color neck
11     andalso all_same_color(neck::tail)
12   | _ => true
13

```

- Give at most a 4 for not having a single case expression with a nested pattern like `head::neck::tail` (though they can use other variable names). But if you see a more complicated pattern like `head::(rest as neck::tail)` do not penalize it. (It technically uses a feature we did not see, but it is arguably better than the sample solution.)
- Do not penalize solutions that use longer patterns for the one-element list than `[_]`. So all of these are also fine:

`[x]` `_::[]` `x::[]` `_::nil` `x::nil`

Remember that you are grading on general style, not how close to the sample solution a student solution is. It is perfectly fine for a solution to be significantly different from the sample, as long as it has good style.

Problem 2e

Here is a sample solution:

```

1 fun sum_cards cs =
2   let fun loop (acc,cs) =
3     case cs of
4       [] => acc
5     | c::cs' => loop (acc + card_value c, cs')
6   in
7     loop (0,cs)
8   end
9

```

As with problem 1c, our focus is on making sure there is a tail-recursive locally defined helper function:

- Give at most a 2 if they do not define a helper function.
- Give at most a 3 if their helper function is not defined locally (inside `sum_cards`).
- Give at most a 3 if the helper function is not tail-recursive: where it calls itself (only one place is needed), there should be no more work afterward. So the result of the recursive call should not be an argument to anything like another function, `+`, etc.
- Give at most a 3 if you cannot easily find the initial call to the helper function with 0 for an accumulator argument.

Other things:

- Give at most a 4 if the solution does not use `card_value` as a helper function.

Remember that you are grading on general style, not how close to the sample solution a student solution is. It is perfectly fine for a solution to be significantly different from the sample, as long as it has good style.

Problem 2f

Here is a sample solution:

```
1 fun score (cs,goal) =
2   let
3     val sum = sum_cards cs
4   in
5     (if sum >= goal then 3 * (sum - goal) else goal - sum)
6     div (if all_same_color cs then 2 else 1)
7   end
```

This problem has an especially large number of reasonable solutions that would be good style. So use your best judgment. Do not necessarily penalize solutions that use conditional expressions less cleverly than the sample solution. However, give at most a 4 if `sum_cards` can be called more than once when the body of `score` is evaluated.

Problem 2g

Here is a sample solution:

```
1 fun officiate (cards,plays,goal) =
2   let
3     fun loop (current_cards,cards_left,plays_left) =
4       case plays_left of
5         [] => score(current_cards,goal)
6       | (Discard c)::tail =>
7         loop (remove_card(current_cards,c,IllegalMove),cards_left,tail)
8       | Draw::tail =>
9         (* note: must score immediately if go over goal! *)
10        case cards_left of
11          [] => score(current_cards,goal)
12        | c::rest => if sum_cards (c::current_cards) > goal
13                     then score(c::current_cards,goal)
14                     else loop (c::current_cards,rest,tail)
15     in
16       loop ([],cards,plays)
17   end
```

For a longer function like this, we cannot expect solutions to look much like the sample solution, so let's list some things to look for:

- A local helper function keeps track of remaining cards, moves, and held cards in some way
- Use of previously defined functions: `remove_card`, `sum_cards`, `score`
- Good use of case-expressions and pattern-matching. Do not require the nested patterns in the outer case expression above where we have different patterns for the head of the list -- a nested case expression is okay instead.
- A moderately longer solution is okay, but a solution twice as long (in terms of amount of code -- do not worry about how much is put on each line) is probably doing something unnecessary.

Use the general scoring guidelines: 5 for a great solution, 4 for one with a small number of issues, etc.

Problems 3a and 3b

You do not need to provide feedback on problems 3a and 3b (the challenge problems), but you are welcome to give text feedback on these problems here if you wish.

Mark as completed

